Visualizing Discrete Dynamical Systems

Thomas W. Polaski
Winthrop University
polaskit@winthrop.edu

Introduction

A dynamical system that evolves according to the difference equation

$$\boldsymbol{x}_{k+1} = A\,\boldsymbol{x}_k$$

where A is a square matrix is called a **discrete linear homogeneous dynamical system**. Such systems are applicable in ecology, medicine, and engineering. They also provide a discrete analogue of the study of systems of differential equations.

We wish to study the long-term behavior of these systems and to determine whether this behavior depends upon the initial state $\boldsymbol{x}_0$. It turns out that the behavior of these systems depends on the eigenvalues and eigenvectors of A.

We begin by studying the **trajectories** of the system using different initial states $\boldsymbol{x}_0$.

Numerical trajectories for $A = \begin{pmatrix} \frac{4}{9} & \frac{1}{9} \\ \frac{1}{18} & \frac{7}{18} \end{pmatrix}$ and a single given initial vector x_0

In[8]:=
```
Clear[n];
maxn = 5;
A = {{4/9, 1/9}, {1/18, 7/18}};
x = {0, 1};
Table[MatrixPower[A, k].x, {k, 0, maxn}] // N // TableForm
```

Graphical trajectories for $A = \begin{pmatrix} \frac{4}{9} & \frac{1}{9} \\ \frac{1}{18} & \frac{7}{18} \end{pmatrix}$ and a single given initial vector x_0

We now use *Mathematica* to **plot** the trajectories of the system using different initial states $\boldsymbol{x}_0$.

In[162]:=
```
Clear[n];
Clear[x];
maxn = 5;
A = {{4/9, 1/9}, {1/18, 7/18}};
x = {0, 1};
plotmax = Max[Flatten[Abs[Transpose[Table[MatrixPower[A, k].x, {k, 0, maxn}]]]]];
Manipulate[Show[Graphics[{PointSize[Large], Point /@ Table[MatrixPower[A, k].x, {k, 0, n}]}],
  Graphics[{Red, Arrow[Table[MatrixPower[A, k].x, {k, 0, n}]]}],
  Axes → True, AspectRatio → 1,
  PlotRange → {{-plotmax, plotmax}, {-plotmax, plotmax}}], {n, 0, maxn, 1}]
```

Graphical trajectories for $A = \begin{pmatrix} \frac{4}{9} & \frac{1}{9} \\ \frac{1}{18} & \frac{7}{18} \end{pmatrix}$ and a set of initial vectors

Trajectories for several different initial states can be plotted simulataneously. For example, points on a square about the origin could be used.

In[20]:=
```
Clear[n];
Clear[x];
maxn = 5;
A = {{4/9, 1/9}, {1/18, 7/18}};
plotmax = Max[Flatten[
    Abs[Transpose[Table[MatrixPower[A, k].{i, j}, {i, -1, 1}, {j, -1, 1}, {k, 0, maxn}]]]]];
Manipulate[Show[Graphics[{PointSize[Large],
    Point /@ Flatten[Table[MatrixPower[A, k].{i, j}, {i, -1, 1}, {j, -1, 1}, {k, 0, n}], 2]}],
  Graphics[{Thick, Red, Table[Arrow[Table[MatrixPower[A, k].{i, j}, {k, 0, n}]],
     {i, -1, 1}, {j, -1, 1}]}], Axes → True, AspectRatio → 1,
  PlotRange → {{-plotmax, plotmax}, {-plotmax, plotmax}}], {n, 0, maxn, 1}]
```

Eigenvalues and eigenvectors

It appears that all trajectories approach the origin. Most spiral in toward the origin, but those that start along two particular lines approach the origin directly. In these cases, the distance from the origin seems to multiplied by a constant λ at each time step; that is,

$$\boldsymbol{x}_{k+1} = A\,\boldsymbol{x}_k = \lambda\,\boldsymbol{x}_k$$

In[33]:=
```
Eigenvalues[A]
```

In[34]:=
```
Eigenvectors[A]
```

Notes:

1. All trajectories tend toward the origin. The origin is an **attractor** of the dynamical system.

2. The origin is an attractor when all eigenvalues have magnitide less than 1.

3. The **direction of greatest attraction** is the line through the origin and the eigenvector $\boldsymbol{v}$ corresponding to the eigenvalue of smallest magnitude.

A helpful coordinate system

The spiraling of the trajectories can be examined by looking at the coordinates of the iterates with respect to a basis of eigenvectors. In less fancy terms, we produce “graph paper” using lines through the origin and the eigenvectors as new coordinate axes.

In[150]:=
```
Clear[n];
maxn = 5;
A = {{4/9, 1/9}, {1/18, 7/18}};
v1 = Eigenvectors[A][[1]];
v2 = Eigenvectors[A][[2]];
g1 = ParametricPlot[Table[k * v2 + t * v1, {k, -10, 10}], {t, -10, 10}, PlotStyle → Blue];
g2 = ParametricPlot[Table[k * v1 + t * v2, {k, -10, 10}], {t, -10, 10}, PlotStyle → Purple];
g3 = ParametricPlot[t * v1, {t, -10, 10}, PlotStyle → {Blue, Thick}];
g4 = ParametricPlot[t * v2, {t, -10, 10}, PlotStyle → {Purple, Thick}];
x = {-1, -2};
plotmax = Max[Flatten[Abs[Transpose[Table[MatrixPower[A, k].x, {k, 0, maxn}]]]]];
Manipulate[Show[Graphics[{PointSize[Large], Point /@ Table[MatrixPower[A, k].x, {k, 0, n}]}],
  Graphics[{Red, Arrow[Table[MatrixPower[A, k].x, {k, 0, n}]]}],
  g1, g2, g3, g4, Axes → True, AspectRatio → 1,
  PlotRange → {{-plotmax, plotmax}, {-plotmax, plotmax}}], {n, 0, maxn, 1}]
```

Trajectories for given eigenvalues and eigenvectors and a set of initial vectors

We can plot the trajectories of a linear dynamical system with eigenvalues λ_1 and λ_2 and corresponding eigenvectors $\boldsymbol{v}_1$ and $\boldsymbol{v}_2$ by letting

$$A = P\,D\,P^{-1}$$

where

$$D = \begin{pmatrix} \lambda_1 & 0 \\ 0 & \lambda_2 \end{pmatrix} \text{ and } P = \begin{pmatrix} \boldsymbol{v}_1 & \boldsymbol{v}_2 \end{pmatrix}$$

A random invertible P can be generated if particular eigenvectors are not important.

- **Example:** $\lambda_1 = 3/2,\ \lambda_2 = 4/3,\ \boldsymbol{v}_1 = \begin{pmatrix} 2 \\ 1 \end{pmatrix},\ \boldsymbol{v}_2 = \begin{pmatrix} -1 \\ 1 \end{pmatrix}$

In[47]:=
```
λ1 = 3 / 2;
λ2 = 4 / 3;
v1 = {2, 1};
v2 = {-1, 1};
x = {1, 0};
Clear[n];
maxn = 5;
P = Transpose[{v1, v2}];
A = P.DiagonalMatrix[{λ1, λ2}].Inverse[P];
A // MatrixForm
Table[MatrixPower[A, k].x, {k, 0, maxn}];
plotmax = Max[Flatten[
    Abs[Transpose[Table[MatrixPower[A, k].{i, j}, {i, -1, 1}, {j, -1, 1}, {k, 0, maxn}]]]]];
Manipulate[Show[Graphics[{PointSize[Large],
    Point /@ Flatten[Table[MatrixPower[A, k].{i, j}, {i, -1, 1}, {j, -1, 1}, {k, 0, n}], 2]}],
  Graphics[{Thick, Red, Table[Arrow[Table[MatrixPower[A, k].{i, j}, {k, 0, n}]],
     {i, -1, 1}, {j, -1, 1}]}], Axes → True, AspectRatio → 1,
  PlotRange → {{-plotmax, plotmax}, {-plotmax, plotmax}}], {n, 0, maxn, 1}]
```

Notes:

1. All trajectories (except for the trajectory starting at the origin) tend away from the origin. The origin is an **repellor** of the dynamical system.

2. The origin is an repellor when all eigenvalues have magnitide greater than 1.

3. The **direction of greatest repulsion** is the line through the origin and the eigenvector ***v*** corresponding to the eigenvalue of largest magnitude.

- **Example:** $\lambda_1 = 1.1$, $\lambda_2 = .9$, $v_1 = \begin{pmatrix} 2 \\ 1 \end{pmatrix}$, $v_2 = \begin{pmatrix} -1 \\ 1 \end{pmatrix}$

In[60]:=
```
λ1 = 1.1;
λ2 = .9;
v1 = {2, 1};
v2 = {-1, 1};
x = {1, 0};
Clear[n];
maxn = 10;
P = Transpose[{v1, v2}];
A = P.DiagonalMatrix[{λ1, λ2}].Inverse[P];
A // MatrixForm
Table[MatrixPower[A, k].x, {k, 0, maxn}];
plotmax = Max[Flatten[
    Abs[Transpose[Table[MatrixPower[A, k].{i, j}, {i, -1, 1}, {j, -1, 1}, {k, 0, maxn}]]]]];
Manipulate[Show[Graphics[{PointSize[Large],
    Point /@ Flatten[Table[MatrixPower[A, k].{i, j}, {i, -1, 1}, {j, -1, 1}, {k, 0, n}], 2]}],
  Graphics[{Thick, Red, Table[Arrow[Table[MatrixPower[A, k].{i, j}, {k, 0, n}]],
     {i, -1, 1}, {j, -1, 1}]}], Axes → True, AspectRatio → 1,
  PlotRange → {{-plotmax, plotmax}, {-plotmax, plotmax}}], {n, 0, maxn, 1}]
```

Notes:

1. Some trajectories tend away from the origin while others tend toward the origin. The origin is a **saddle point** of the dynamical system.

2. The origin is a saddle point when some eigenvalues are greater than 1 in magnitude and others are smaller than 1 in magnitude.

3. The **direction of greatest repulsion** is the line through the origin and the eigenvector ***v*** corresponding to the eigenvalue of largest magnitude. The **direction of greatest attraction** is the line through the origin and the eigenvector ***v*** corresponding to the eigenvalue of smallest magnitude.

Trajectories for given eigenvalues and random eigenvectors, and a set of initial vectors

- **Example:** $\lambda_1 = 1/2$, $\lambda_2 = 1/3$, **random eigenvectors**

In[133]:=
```
λ1 = 1 / 2;
λ2 = 1 / 3;
Clear[n];
maxn = 5;
P = Table[RandomReal[], {i, 2}, {j, 2}]
A = P.DiagonalMatrix[{λ1, λ2}].Inverse[P];
A // MatrixForm
Table[MatrixPower[A, k].x, {k, 0, maxn}];
plotmax = Max[Flatten[
    Abs[Transpose[Table[MatrixPower[A, k].{i, j}, {i, -1, 1}, {j, -1, 1}, {k, 0, maxn}]]]]]];
Manipulate[Show[Graphics[{PointSize[Large],
    Point /@ Flatten[Table[MatrixPower[A, k].{i, j}, {i, -1, 1}, {j, -1, 1}, {k, 0, n}], 2]}],
  Graphics[{Thick, Red, Table[Arrow[Table[MatrixPower[A, k].{i, j}, {k, 0, n}]],
     {i, -1, 1}, {j, -1, 1}]}], Axes → True, AspectRatio → 1,
  PlotRange → {{-plotmax, plotmax}, {-plotmax, plotmax}}], {n, 0, maxn, 1}]
```